Flowing text 
(Removing paragraph marks from the end of each line)
Plain text files are often formatted with each line as a separate paragraph. The FlowText macro will flow the text into continuous paragraphs. Requirements:

There must be two successive paragraph marks at points where you want the text divided into actual paragraphs, i.e. actual paragraphs must be separated by an empty paragraph. The empty paragraphs will be removed when the macro runs. 
For best results, turn on visibility of formatting characters:

· Word 2007: Office button, Word Options, Display, check Show all formatting marks

· Previous versions: Tools > Options > View > Formatting Marks, check ALL.
To use: If no text is selected, text is flowed from the current cursor position to the end of the document. 
If there is a selection, only the selection is flowed. This is useful, for example, in cases where a portion of the text should remain in separate single-line paragraphs. 
CAUTION: The macro also replaces all line-ends () with paragraphs marks (¶) – useful for flowing text downloaded from web pages. If you don’t want this behavior, delete or comment out the first Find/Replace execution, as indicated in the code below. 
Copy and paste the VBA code below into a new macro, either in Normal.dot or some other .dot file. I have assigned it to the keyboard shortcut Alt-F for quick access. 
Sub FlowText()

' Author: Ted Mirecki

' Initially created 01/2002, this version 04/2008
    With Selection.Find

        .ClearFormatting

        .Replacement.ClearFormatting

        .Forward = True

        .Wrap = wdFindStop

        .Format = False

        .MatchCase = False

        .MatchWholeWord = False

        .MatchWildcards = False

        .MatchSoundsLike = False

        .MatchAllWordForms = False

    End With

' Replace line-ends with parag marks
' If this is undesirable, delete or comment out next statement
    Selection.Find.Execute FindText:="^l", ReplaceWith:="^p", _
 
Replace:=wdReplaceAll

' Delete any white space in front of parag mark
    Selection.Find.Execute FindText:="^w^p",ReplaceWith:="^p", _
 
Replace:=wdReplaceAll

' Mark end of parag with special marker
    Selection.Find.Execute FindText:="^p^p",ReplaceWith:="qq", _
 
Replace:=wdReplaceAll

' Replace each parag mark with space
' This does the actual text flow
    Selection.Find.Execute FindText:="^p", ReplaceWith:=" ", _
 
Replace:=wdReplaceAll

' Replace markers with actual parag marks
    Selection.Find.Execute FindText:="qq", ReplaceWith:="^p", _
 
Replace:=wdReplaceAll
' Note: If parags are to be separated by an empty one,
' use a ReplaceWith string of "^p^p"
End Sub
